René 6427

HP-97 Computer Programs for use in Benthic Ecological Research

Lesley McGhee and D. J. Wildish

Biological Station,
St. Andrews, N. B., EOG 2XO

December 1983

Canadian Manuscript Report of Fisheries and Aquatic Sciences No. 1742

Government of Canada Fisheries and Oceans
Gouvernement du Canada Pêches et Océans

Canadian Manuscript Report of
Fisheries and Aquatic Sciences 1742

December 1983

HP-97 COMPUTER PROGRAMS FOR USE IN BENTHIC ECOLOGICAL RESEARCH

by

Lesley McGhee and D. J. Wildish

Fisheries and Environmental Sciences
Fisheries Research Branch
Department of Fisheries and Oceans
Biological Station
St. Andrews, New Brunswick EOG 2XO
Canada

© Minister of Supply and Services Canada 1983

Cat. No. Fs 97-4/1742 ISSN 0706-6473

Correct citation for this publication:

McGhee, Lesley, and D. J. Wildish. 1983. HP-97 computer programs for use in benthic ecological research. Can. MS Rep. Fish. Aquat. Sci. 1742: iv + 27 p.

CONTENTS

ABSTRACT

McGhee, Lesley, and D. J. Wildish. 1983. *HP-97 computer programs for use in benthic ecological research.* Can. MS Rep. Fish. Aquat. Sci. 1742: iv + 27 p.

Nine programs for use in ecological research, particularly in benthic ecology, are presented. Each program is recorded on a magnetic card compatible with the HP-97 and is available from the Users' Library. Four new programs have been written and full documentation for each is recorded in this report. In addition, two modifications of existing programs (ATP analysis and growth) and an aide memoire for production calculations are presented.

Key words: Microcomputer programs, benthic ecology

RÉSUMÉ

McGhee, Lesley, and D. J. Wildish. 1983. HP-97 computer programs for use in benthic ecological research. Can. MS Rep. Fish. Aquat. Sci. 1742: iv + 27 p.

Le présent rapport contient neuf programmes pouvant être utilisés dans des recherches écologiques, en particulier en écologie benthique. Chaque programme est enregistré sur une carte magnétique convenant à l'ordinateur HP-97 et qu'on peut se procurer de la magnétothèque des utilisateurs. Quatre nouveaux programmes ont été rédigés et nous donnons ici la documentation complète pour chacun de ces programmes. Nous présentons de plus deux modifications apportées à des programmes existants (analyse ATP et croissance) et un aide-mémoire pour des calculs de production.

INTRODUCTION

The need to analyze field collected data, experimental results, or chemical/physical measurements frequently involves the use of complicated or tedious mathematical computations which are prone to errors when performed by hand. The desk-top calculator manufactured by Hewlett-Packard (HP-97 or HP-67) provides an economical means of avoiding tediousness and improving the accuracy of analytical computations.

The HP-97 has a printer which provides hard copy on pressure sensitive paper and a program reader capable of storing a computer program prerecorded on a magnetic card by making the appropriate series of key strokes on the calculator. The magnetic card programs can readily be copied by the HP-97. In addition, Hewlett-Packard runs a Users' Library of contributed programs.

It is the purpose here to list the programs available from the Users' Library which are of potential use in aquatic, particularly benthic, ecological research, and to fully document programs written by us for this purpose. The number preceding every program listed is the order number and by reference to it the program can be obtained from: Users' Library, Hewlett-Packard Co., Dept. 39VL, 1000 NE Circle Blvd., Corvallis, OR 97330, U.S.A.

PROGRAM LISTING

97-02019-8 Growth equation fitted by iterative least squares method.

Fits the equation $y = a/(1+\exp(-k*(kx0)))$

where x = time
y = size

A maximum of 20 x,y data points is permitted; finds the value of a, k and x0.

97-02745-8 Automatic selection of best curve fit.

Data can be treated in one of two ways:

1. Selection of one (of four) types of curve fitting and printing out of the coefficient of determination;

2. Selection of all four types of curve fitting (linear, exponential, logarithmic and power curve) and printing out of the coefficient of determination for each; curve selection is then by magnitude of the value of the correlation coefficient which is ranked for each fitting by this program.

97-02752-4 An ecological model.

This program solves the differential equation:

$$\frac{dy}{dx} = \frac{y(cx-p)}{x(a-by)}$$

y = population of the predators
x = population of the prey
a, b, c, p = positive constants

97-03639-2 Population density estimate.

The population considered is assumed to be randomly dispersed. A linear transect through the midst of the animals' range is made with the observer recording distance to each animal seen. Population density as number/unit area ± standard error is calculated after input of a constant which is characteristic for the species concerned and accounts for species differences in visibility.

97-03796-0 Margalef information theory diversity index.

97-03809-1 Estimate of association for species distributions.

97-03936-2 Brillouin's diversity index (Hmax, Hmin, V).

Equitability (V) is based on the ratio of H to Hmax and Hmin.

97-04393-5 Population growth.

Given a constant growth rate, this growth model can solve for any one of four variables (initial size, final size, doubling time, time to final size) given the other three.

97-04694-6 Comparative diversity indices.

The following are included: Shannon's, Brillouin's, Simpson's, Pielou's modification of Simpson's and McIntosh's diversity indices. An evenness index based on the Shannon formula is also given.

NEW PROGRAMS

PRODUCTION BY COHORT SUMMATION OF LOSSES METHOD

This program calculates average standing crop, B, production, P, and turnover ratio, P:B, from a temporal series of samples and from which discrete cohorts can readily be recognized. Input data are the densities, N, and average weights, w, of an individual. The following calculations are made for each time interval, except the first:

$$B = N_w$$

$$P = \sum_{t=0}^{t} \sum_{0}^{N} G_w \Delta t$$

where t is the maximum survivorship time and G is the mean instantaneous relative growth of an individual, and integrated growth for a given time interval is given by

$$G\Delta t = \frac{1}{w}\left(\frac{dw}{dt}\right)\Delta t.$$

Running totals of B, P and P:B are stored and printed after every pair of entries. After the last entry, P is corrected by:

$$P_{cohort} = P \times CPI$$

where CPI = 12/survivorship time in months and the cohort turnover ratio determined as Pcohort:B.

Annual production depends on the number of cohorts present during a year and hence on information not used in this program which pertains to a single, named cohort.

Sample problem

Data for this example were taken from Wildish and Peer (1981). This reference also describes the computations needed for application of the cohort method to marine amphipods.

Given the date, density, and mean weight per animal, find the standing crop, production per sample and the sample P:B. From these calculate the total production, total standing crop, cohort production interval, annual production, mean biomass and the overall P:B ratio.

Sampling date	Density no./m^2	Mean weight per animal (mg)
May 1971	19500	0.01
June	4200	0.1
July	1600	0.8
August	900	2.5
September	620	4.5
October	540	5.6
November	502	6.6
December	502	7.3
January 1972	350	8.4
February	250	10.8
March	235	13.4
April	215	16.5
May	190	19.2
June	35	20.7

	Input	Function	Display	Comments
		A	PRINTED	Start program
	571	R/S	571	Enter date (5th month 1971)
	19500	R/S	19500	Enter density
	0.01	R/S	0.01	Enter mean weight
			195	1st standing crop
(1)				
	6	R/S	6	Enter new date
	4200	R/S	4200	Enter new density
	0.1	R/S	0.1	Enter new mean weight
			420	Standing crop for the sample
			1066.5	Production for the sample
			1.73415	P:B

Return to (1) until all data are entered.

-5	R/S		Any negative number will work
		33645.8	Total standing crop
		11484	Total production
12	R/S	1	Cohort production interval
		11484	Annual production
		2403.27143	Mean biomass
		4.77849	P:B

	INSTRUCTIONS	INPUT	FUNCTIONS		DISPLAY
1	LOAD THE PROGRAM				PRINTED
2	START THE PROGRAM			A	
3	ENTER DATE	DATE		R/S	DATE
4	ENTER FIRST DENSITY	No		R/S	No
5	ENTER FIRST MEAN WEIGHT TO GET SAMPLE BIOMASS(B)	Wo		R/S	Wo B
6	ENTER NEXT DATE	NDATE		R/S	NDATE
7	ENTER NEXT DENSITY	N1		R/S	N1
8	ENTER NEXT WEIGHT	W1		R/S	W1 B PRODUCTION SAMPLE P:B
9	RETURN TO 6 UNTIL ALL THE DATA ARE ENTERED				
10	ENTER ANY NEGATIVE NUMBER TO GET THE TOTAL STANDING CROP (TSC) AND THE TOTAL PRODUCTION(TP)	NUM		R/S	TSC TP
11	ENTER THE TIME PERIOD eg. NUMBER OF MONTHS TO GET THE COHORT PRODUCTION INTERVAL, ANNUAL PRODUCTION, AVERAGE BIOMASS, AND OVERALL P:B	TIME		R/S	CPI Pannual AVE B P:B
	AN ERROR IN INPUT CAN ONLY BE CHANGED RIGHT AFTER THE ERROR IS PRINTED, BEFORE THE NEXT NUMBER IS ENTERED				
	TO CHANGE DENSITY	N		D R/S	DATE N
	TO CHANGE THE DATE	NDATE		C R/S	NDATE
	THE MEAN WEIGHT CANNOT BE CHANGED				
	TO RESTART THE PROGRAM AND CLEAR ALL REGISTERS			A	

PROGRAM LISTING

```
001 *LBLA          LBLA-LBLC INITIALIZE REGISTERS
002  CLRG
003  1             SET UP A COUNT OF SAMPLES
004  STO0
005  DSP0
006  R/S           ENTER AND PRINT THE FIRST DATE.
007  PRTX
008  STO5          STORE IT IN REGISTER 5.
009  DSP5
010 *LBLc
011  R/S           ENTER THE FIRST DENSITY (No) AND
012  STOA          STORE IT IN REGISTER A.
013  PRTX
014 *LBLe
015  R/S           ENTER THE FIRST MEAN WEIGHT (Wo) AND
016  STOB          STORE IT IN REGISTER B.
017  PRTX
018  x
019  STO4          No x Wo = STANDING CROP (Bo)
020  PRTX
021 *LBLa
022  R/S
023  SPC
024  X<0?          CHECK TO SEE IF THE INPUT IS COMPLETE
025  GTOB          (A NEGATIVE NUMBER ENTERED HERE WILL
026  STO5          STOP THE PROGRAM) ENTER THE NEW DATE.
027  DSP0
028  PRTX
029  DSP5
030  1
031  ST+0          KEEP A RUNNING TOTAL OF THE NUMBER OF
032 *LBLd          SAMPLES IN REGISTER 0.
033  R/S           ENTER A NEW DENSITY (N1).
034  STO2
035  PRTX
036 *LBLE
037  R/S           ENTER THE NEW MEAN WEIGHT (W1).
038  STO3
039  PRTX
040  x             N1 x W1 = B1
041  PRTX
042  ST+4          ADD THE NEW STANDING CROP TO TOTAL
043  RCLA
044  RCL2
045  +
046  2             (No+N1)/2 x (Wo-W1)=PRODUCTION
047(SORRY NO SYMBOL FOR DIVIDE IS AVAILABLE)
048  RCL3
049  RCLB          No=FIRST DENSITY  Wo=FIRST MEAN WEIGHT
050  -             N1=NEXT DENSITY    W1=NEXT MEAN WEIGHT
051  x
052  PRTX
```

```
053  ST+7          KEEP A RUNNING TOTAL OF THE PRODUCTION
054  RCL4
055?DIVIDE         PRODUCTION/CURRENT TOTAL BIOMASS=SAMPLE P:B
056  PRTX
057  RCL2
058  STOA          SWITCH W1 TO Wo
059  RCL3          AND N1 TO No
060  STOB
061  GTOa          RETURN FOR NEW INPUT
062 *LBLB          LBLB-LBLD FOR PRINTING FINAL VALUES
063  SPC
064  RCL4          PRINT TOTAL STANDING CROP (BIOMASS)
065  PRTX
066  RCL7          PRINT TOTAL PRODUCTION
067  PRTX
068  1
069  2
070  ENT           12/NUMBER OF MONTHS=COHORT  PRODUCTION
071  R/S                       INTERVAL (CPI)
072?DIVIDE
073  PRTX
074  RCL7
075  x             TOTAL PRODUCTIONxCPI=ANNUAL PRODUCTION (Pannual)
076  PRTX
077  RCL4
078  RCL0
079?DIVIDE         TOTAL BIOMASS/NUMBER OF SAMPLES=AVERAGE
080  PRTX                                    BIOMASS
081?DIVIDE         Pannual/AVERAGE BIOMASS=P:B
082  PRTX
083  RTN
084 *LBLD
085  DSP0
086  SPC
087  RCL5          CORRECT THE DENSITY VALUE
088  PRTX
089  DSP5
090  RCL0
091  1
092  X=Y?          SPECIAL CONDITIONS FOR THE FIRST SAMPLE
093  GTOc
094  GTOd
095 *LBLC
096  SPC
097  1
098  ST-0
099  RCL0          CORRECTING THE FIRST SAMPLE
100  0
101  X=Y?
102  GTOA
103  GTOa
104  RTN
```

PRODUCTION BY SIZE-FREQUENCY METHOD

This program calculates average standing crop, B, production, P, and turnover ratio, P:B, from a single ranked series of samples taken over a period of time and for which discrete cohorts cannot be recognized. Input data are the chosen number of size groups, i, the density, N_j, and W_j, the mean weight in the jth size group. Running totals of P, B and P:B are printed and the final P and P:B value corrected by an inputted cohort production interval (CPI).

Calculations made are:

$$B = \sum_{j=1}^{i} N_j(W_j)$$

$$P = i \sum_{j=1}^{i} (N_j - N_{j-1})\left(\frac{W_j + W_{j+1}}{2}\right).$$

The final P value is corrected as follows:

$$P_{annual} = P \times CPI$$

where CPI = 12/survivorship time in months and the annual turnover ratio found by

$$P_{annual}:B.$$

Because size class, i, is the last containing a density value $N_{i+1} = 0$, W_{i+1} must be estimated in order to calculate

$$\frac{W_i + W_{i+1}}{2}.$$

Sample problem

The data for this example were taken from Wildish and Peer (1981). The total standing crop, total production, annual production and P:B ratio from the density and mean biomass of each size group are given below:

Density	Mean biomass (mg)
45	1.25
61.8	3
71.1	5.75
23.3	10
22.8	16.25
9.8	26.25
2.9	37.5
0.4	50

Use 0 as the last density and 54 as the last biomass. The number of losses is 8.

	Input	Function	Display	Comments
		A	PRINTED	Start program
	8	R/S		Enter number of losses
	45	R/S		Enter first density (N_{j-1})
	1.25	R/S		Enter first mean weight (W_{j-1})
			1	Sample number
			45	N_{j-1}
			1.25	W_{j-1}
			56.25	Standing crop for the sample
(1)	61.8	R/S		Enter next density (N_j)
	3	R/S		Enter next biomass (W_j)
			-285.6	Production for sample
			2	Sample number
			61.8	N_j
			3	W_j
			185.4	Standing crop

Return to (1) until all data are entered.

		1639.975	Total standing crop
		7463.7	Total production
15	R/S		Enter number of months
		0.8	Production interval
		5970.96	Annual production
		3.6409	Pannual:B ratio

If the density or biomass are entered incorrectly, they can be changed:

Input	Function	Display	Comments
45	R/S		Enter an incorrect density
1.3	R/S		Enter a correct biomass
		4990	Production for previous sample (incorrect)
		5	Sample number (correct)
		45	N_j (incorrect)
		1.3	W_j (correct)
		58.5	Standing crop (incorrect)
	D		To change density
4.5	R/S		Enter correct density
		4	Previous sample number
		489	Production (correct)
		5	Sample number (correct)
		4.5	N_j (correct)
		1.3	W_j (correct)
		5.85	Standing crop (correct)
4.5	R/S		Enter a correct density
2.5	R/S		Enter an incorrect biomass
		4000	Production for previous sample (incorrect)
		5	Sample number (correct)
		4.5	N_j (correct)
		2.5	W_j (incorrect)
		11.25	Standing crop (incorrect)
	C		To change biomass
1.3	R/S		Enter correct biomass
		4	Previous sample number
		489	Production (correct)
		5	Sample number (correct)
		4.5	N_j (correct)
		1.3	W_j (correct)
		5.85	Standing crop (correct)

	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	LOAD THE PROGRAM			PRINTED
2	START THE PROGRAM		A	
3	ENTER NO. OF TIMES LOSSES OCCURRED	LOSSES	R/S	
4	ENTER DENSITY	Nj-1	R/S	
5	ENTER MEAN BIOMASS	Wj-1	R/S	
	THIS WILL CAUSE THE PRINTING OF THE SAMPLE NUMBER, THE FIRST DENSITY, THE FIRST MEAN BIOMASS, AND THE STANDING CROP FOR THE SAMPLE (S C).			SAMPLE # Nj-1 Wj-1 S C
6	ENTER NEXT DENSITY	Nj	R/S	
	ENTER NEXT MEAN BIOMASS THIS WILL GIVE THE SAMPLE PRODUCTION, THE Nj AND Wj BECOME Nj-1 AND Wj-1 . THE DISPLAY FROM STEP 5 IS GIVEN	Wj	R/S	PRODUCTION
7	RETURN TO STEP 6 UNTIL ALL THE DATA ARE ENTERED. WHEN ALL THE DATA ARE ENTERED THE PROGRAM WILL PRINT THE TOTAL STANDING CROP (TSC) AND THE TOTAL PRODUCTION (TP)			TSC TP
8	ENTER THE NUMBER OF MONTHS THIS WILL GIVE THE THE PRODUCTION INTERVAL (P I), THE ANNUAL PRODUCTION (Pannual), AND THE P:B RATIO.	TIME	R/S	P I Pannual P:B
	AN ERROR IN INPUT CAN ONLY BE CHANGED RIGHT AFTER IT IS PRINTED, BEFORE THE NEW DENSITY AND BIOMASS ARE ENTERED TO CHANGE THE DENSITY.		D	
		Nj	R/S	LAST SAMPLE# PRODUCTION SAMPLE# Nj Wj S C
	RETURN TO STEP 6.			
	TO CHANGE THE BIOMASS.		C	
		Wj	R/S	LAST SAMPLE# PRODUCTION SAMPLE# Nj Wj S C
	RETURN TO STEP 6.			

PROGRAM LISTING

```
001 *LBLA          INITIALIZATION SECTION LBLA -LBLa
002  CLRG          CLEAR REGISTERS
003  R/S           STOP FOR INPUT OF THE # OF LOSSES
004  DSP5          SET TO 5 DECIMAL PLACES
005  STO8          STORE # OF LOSSES IN REGISTER 8
006  R/S           STOP FOR FIRST DENSITY (Nj-1)
007  STO2          STORE IN REGISTER 2
008  R/S           STOP FOR FIRST MEAN BIOMASS (Wj-1)
009  STO4          STORE IN REGISTER 4
010 *LBLa
011  SPC
012  DSP0
013  1             KEEP A COUNT OF THE SAMPLES IN REGISTER 0
014  ST+0
015  RCL8
016  RCL0          THE PROGRAM IS FINISHED WHEN THE NUMBER
017  X>Y?          OF SETS OF DENSITIES IS ONE MORE THAN
018  GTOB          THE NUMBER OF LOSSES.
019  RCL0
020  PRTX
021  DSP5
022  RCL2
023  PRTX          PRINT THE DENSITIES (Nj-1) AND THE MEAN
024  RCL4          BIOMASSES (Wj-1).
025  PRTX
026  x             Nj-1 x Wj-1 = STANDING CROP (S C)
027  PRTX
028  ST+5          KEEP A RUNNING TOTAL OF THE S C
029  R/S           ENTER THE NEXT DENSITY (Nj)
030  STOC
031  RCL2
032  STOA
033  RCL4          CHANGE Nj TO Nj-1 BY SWITCHING
034  STOB          REGISTERS.
035  RCLC
036  STO2
037  R/S           STOP FOR Wj
038  STO4
039  RCLB
040  RCL4
041  +
042  2
043?DIVIDE
044  RCLA        (Wj-1 + Wj)/2 x (Nj-1 - Nj) x # OF LOSSES
045  RCL2              =PRODUCTION FOR THE SAMPLE
046  -
047  x
048  RCL8
049  x
050  PRTX
051  ST+9          KEEP A RUNNING TOTAL OF THE PRODUCTION.
052  GTOa          START AGAIN.
053 *LBLB
054  DSP5
055  RCL5          PRINT STANDING CROP TOTAL (TSC).
056  PRTX
057  RCL9          PRINT PRODUCTION TOTAL (TP).
058  PRTX
059  1
060  2
061  R/S           DIVIDE THE NUMBER OF MONTHS IN A YEAR BY
062?DIVIDE         THE SAMPLING TIME TO GET THE PRODUCTION
063  PRTX          INTERVAL (PI).
064  RCL9          MULTIPLY PI BY THE TP TO GET THE ANNUAL
065  x             PRODUCTION (PA).
066  PRTX
```

```
067  RCL5          DIVIDE Pannual BY THE TOTAL STANDING CROP TO
068?DIVIDE         GET THE P:B RATIO.
069  PRTX
070  RTN
071 *LBLC          LAST SECTION FOR CORRECTING ANY ERRORS IN THE
072  4             LAST INPUT.
073  STOE
074  GTOe          REGISTER E IS USED AS A FLAG TO TELL WHERE THE
075 *LBLD          DENSITY (LBLD-2) OR THE BIOMASS (LBLC-4) IS
076  2             TO BE CHANGED.
077  STOE
078 *LBLe
079  RCL2
080  RCL4          FIND THE INCORRECT STANDING CROP WHICH WAS
081  x             ADDED TO THE TOTAL IN REG 5 AND SUBTRACTS IT.
082  ST-5
083  RCL0
084  1             IF IT IS THE FIRST SAMPLE THAT IS BEING
085  X=Y?          CORRECTED, SUBTRACTING THE PRODUCTION IS NOT
086  GTOd          NECESSARY SINCE TOTAL-R9 IS STILL ZERO ( THE
087  RCLB          FIRST PRODUCTION IS NOT CALCULATED UNTIL
088  RCL4          Nj   AND Wj   ARE ENTERED.)
089  +
090  2
091?DIVIDE
092  RCLA
093  RCL2          SUBTRACT THE PRODUCTION WORKED OUT WITH
094  -             THE WRONG VALUES FROM THE TOTAL.
095  x
096  RCL8
097  x
098  ST-9
099 *LBLd
100  4
101  RCLE
102  X=Y?
103  GTOb
104  R/S
105  STO2          CHECK THE FLAG IN REGISTER E TO FIND OUT
106  GTOE          WHETHER IT IS THE BIOMASS (REGISTER 4) OR
107 *LBLb          THE DENSITY (REGISTER 2) THAT IS TO BE
108  R/S           CHANGED
109  STO4
110 *LBLE
111  1             SUBTRACT 1 FROM THE SAMPLE NUMBER COUNT.
112  ST-0
113  DSP0
114  0
115  RCL0          CHECK TO SEE IF IT IS THE FIRST SAMPLE
116  X=Y?          WHICH HAS BEEN CHANGED. IF SO, THERE IS NO
117  GTOa          NEED TO CHANGE THE PRODUCTION TOTAL.
118  SPC           PRINT PREVIOUS SAMPLE NUMBER.
119  PRTX
120  DSP5
121  RCLB
122  RCL4
123  +
124  2
125?DIVIDE         CALCULATE THE CORRECT PRODUCTION FOR THE
126  RCLA          PREVIOUS SAMPLE, ADD IT TO THE TOTAL (IN
127  RCL2          REGISTER 9) AND PRINT IT.
128  -
129  x
130  RCL8
131  x
132  ST+9
133  PRTX
134  GTOA
135  RTN
```

SPECIFIC GROWTH, RATION AND CONVERSION EFFICIENCY FOR INDIVIDUAL ANIMALS

The purpose of this program is to calculate the parameters of growth for individual animals within an experimental group for a single observational period, t, in days. Input parameters are the initial, W_0, and final, W_1, biomasses, initial, L_0, and final, L_1, lengths and the total ration, R_t, for the period, t, considered. The following calculations are made for each animal:

$$G = [(\log W_1 - \log W_0)/n] \times 100\%$$

where G is the specific growth rate per day;

$$R = [\frac{R_t}{t \times W_0}] \times 100\%$$

where R is the ration per animal per day;

$$GCE = (G/R) \times 100\%$$

where GCE is the gross conversion efficiency;

R_m = value of R when G = 0 in linear relationship of G on R;

$$NCE = [\frac{G}{R - R_m}] \times 100\%$$

where NCE is the net conversion efficiency; and

$$CF = [W/L^3] \times 100\%$$

where CF = condition factor, W = biomass, and L = length

Sample problem

Use the following data:

Initial weight	Final weight	Initial length	Final length
71.2	79.45	18	18.94
69.9	76.01	18.04	18.61
64.5	66.61	17.23	17.61
73.6	77.5	18.2	19.11
59	40.33	17.2	15.971
44.7	58.22	15.42	16.88

Total food = 19.54

Input	Function	Display	Comments
		PRINTED	
	A		Start program
28	R/S		Enter time period
19.54	R/S		Enter total ration
(1) 71.2	R/S		Enter initial weight
			Sample #
		1	
		71.2	W_0
79.45	R/S	79.45	Enter final weight
18.0	R/S		Enter initial length
		18.0000	L_0
		1.22085	Initial condition factor
18.94	R/S		Enter final length
		18.94	L_1
		1.16938	Final condition factor
		0.41382	Growth
		0.98014	Ration
		42.22108	GCE
			Go to (1) until all the data are entered then type [E]
	E		
		1.17927	Average CF
		0.16369	Average growth
		1.12544	Average ration
		12.98192	Average GCE
		0.10425	R^2
		-0.97753	a
		1.01403	b
		0.96401	Maintenance ration
(2) 0.41382	R/S		
0.98014	R/S		
		1.00000	Sample #
		25.65690	Go to (2) until all the NCE'S have been calculated

	INSTRUCTIONS	INPUT	FUNCTION		DISPLAY
1	LOAD THE PROGRAM				PRINTED
2	START THE PROGRAM			A	
3	ENTER EXPERIMENT LENGTH (DAYS).	T		R/S	
4	ENTER TOTAL RATION	TOTR		R/S	
5	ENTER INITIAL WEIGHT	Wo		R/S	
					SAMPLE # Wo
6	ENTER FINAL WEIGHT	W1		R/S	W1
7	ENTER INITIAL LENGTH CAUSING THE INITIAL CONDITION FACTOR TO BE PRINTED	Lo		R/S	Lo CF
8	ENTER FINAL LENGTH THIS CAUSES THE FINAL CONDITION FACTOR TO BE PRINTED IF THERE IS NO INFORMATION ON THE INITIAL OR FINAL LENGTHS TYPE B AND THE SECTION ON THE CF'S WILL BE SKIPPED ENTERING ZERO FOR EITHER THE INITIAL OR FINAL LENGTH IMPLIES THAT THE LENGTH IS MISSING SO NO CF WILL BE WORKED OUT.	L1		R/S	L1 CF GROWTH RATION GCE
9	RETURN TO STEP 5 FOR THE NEXT SAMPLE.				
10	WHEN ALL THE SAMPLES ARE ENTERED... SO YOU GET THE AVERAGE CONDITION FACTOR (ACF), AVERAGE GROWTH (AG), AVERAGE RATION (AR), AVERAGE GCE, CURVE EFFICIENCY (R SQUARED), FIRST LINEAR CONSTANT (a), SECOND LINEAR CONSTANT (b), AND THE MAINTENANCE RATION (MR).			E	ACF AG AR AGCE R SQUARED a b MR
11	ENTER THE GROWTH.	G		R/S	
					SAMPLE #
12	ENTER THE RATION.	R		R/S	NCE
13	RETURN TO STEP 11 UNTIL ALL THE GROWTH AND RATION VALUES HAVE BEEN ENTERED.				

INSTRUCTIONS	INPUT	FUNCTION		DISPLAY
IF AN ERROR IS MADE WHEN TYPING IN THE THE INITIAL OR FINAL WEIGHTS OR LENGTHS WAIT UNTIL THE PROGRAM STOPS FOR THE NEXT INITIAL WEIGHT THEN ... RE-ENTER THE DATA FROM THE SAMPLE YOU WANT TO CORRECT.		F	A	
INITIAL WEIGHT	Wo		R/S	LAST SAMPLE# Wo
FINAL WEIGHT	W1		R/S	W1
INITIAL LENGTH	Lo		R/S	 Lo CF
FINAL LENGTH	L1		R/S	 L1 CF
THIS WILL ERASE ALL RECORD OF THAT ENTRY FROM THE MEMORY OF THE PROGRAM. RETURN TO STEP 5 AND ENTER THE CORRECT SAMPLE AND THE REST OF THE DATA.				
IF AN ERROR OCCURS IN THE ENTRY OF THE GROWTH OR RATION JUST RE-ENTER THE SAMPLE. THE DATA FROM THIS SECTION IS NOT STORED FOR USE ANYWHERE ELSE.				
TO BEGIN AGAIN JUST TYPE ... ALL THE MEMORIES WILL BE ERASED.			A	
TO CHANGE THE TOTAL RATION WAIT UNTIL THE PROGRAM STOPS FOR THE NEXT INITIAL WEIGHT THEN... RETURN TO STEP 4.		F	D	

PROGRAM LISTING

```
001 *LBLA
002  CLRG
003  P=S          CLEAR PRIMARY AND SECONDARY REGISTERS
004  CLRG
005  R/S
006  STO1         STORE THE LENGTH OF THE EXPERIMENT (DAYS)
007  SF1          IN REGISTER 1. SET FLAGS.
008  CF0
009 *LBLd
010  R/S
011  STO5         STORE TOTAL RATION IN REGISTER 5.
012 *LBLC
013  R/S
014  STO2         STORE THE INITIAL WEIGHT (Wo) IN REGISTER 2.
015  DSP0
016  1
017  F1?
018  ST+0         COUNT THE NUMBER OF SAMPLES ENTERED.
019  F0?
020  ST-0
021  SPC
022  RCL0
023  PRTX
024  DSP5
025  RCL2
026  PRTX
027  R/S
028  STO3          STORE THE FINAL WEIGHT (W1) IN REGISTER 3.
029  PRTX
030  RCL2
031  STO4
032  CF2
033 *LBL1
034  R/S
035  X=0?         IF THE LENGTH EQUALS ZERO SKIP THIS SECTION.
036  GTOb
037  SPC
038  PRTX         WEIGHT/(LENGTH CUBED) x 100% =
039  3                          CONDITION FACTOR
040  Y EXPONENT X
041?DIVIDE
042  1
043  0
044  0
045  x
046  PRTX
047  F1?
048  ST+9          KEEP A RUNNING TOTAL OF THE CONDITION
049  F0?           FACTORS (CF)
050  ST-9
051  RCLB
052  1
053  F1?
054  +             COUNT THE NUMBER OF TIMES THE CF IS WORKED
055  F0?           OUT.
056  -
057  STOB
058 *LBLb
059  F2?
060  GTOB
061  RCL3
062  STO4
063  SF2
064  GTO1
065 *LBLB
066  SPC
067  RCL3
068  LOG
069  RCL2
070  LOG
071  -
072  RCL1          GROWTH (G) =((LOG(W1)-LOG(Wo))/TIME) x 100%
073?DIVIDE
```

```
075  0
076  0
077  x
078  STO4
079  PRTX
080  F1?
081  ST+8          KEEP A RUNNING TOTAL OF THE GROWTH IN
082  F0?           REGISTER 8.
083  ST-8
084  RCL5
085  RCL1
086  RCL2          RATION (R)= ((TOTAL RATION/(Wo x TIME))
087  x                         x 100%
088?DIVIDE
089  1
090  0
091  0
092  x
093  F1?
094  ST+7          KEEP A RUNNING TOTAL OF THE RATION IN
095  F0?           REGISTER 7.
096  ST-7
097  PRTX
098  STOA
099  STOD
100  RCL4
101  X=Y           (SWITCH X AND Y, IMPROPER SYMBOL)
102  STOC
103  F1?
104SUMMATION SIGN (NO SYMBOL AVAILABLE)
105  F0?
106NEGATIVE SUMMATION (NO SYMBOL AVAILABLE)
107  ENT           SET UP THE REGISTERS FOR LINEAR REGRESSION
108  1             (BETWEEN GROWTH AND RATION).
109  +
110  RCLC
111  X=Y           (SWITCH X AND Y, IMPROPER SYMBOL)
112  RCLD
113  X=Y           (SWITCH X AND Y, IMPROPER SYMBOL)
114  RCL4
115  RCLA
116?DIVIDE
117  1             GCE=(G/R)x100%
118  0
119  0
120  x
121  F1?
122  ST+6
123  F0?           KEEP A RUNNING TOTAL OF THE GCE IN
124  ST-6          REGISTER 6.
125  SPC
126  PRTX
127  CF0
128  SF1
129  GTOC
130 *LBLE
131  RCL9
132  RCLB
133  X=0?          FIND THE AVERAGE CF.
134?DIVIDE
135  SPC
136  PRTX
137  RCL8
138  RCL0          FIND THE AVERAGE GROWTH.
139?DIVIDE
140  PRTX
141  RCL7
142  RCL0          FIND THE AVERAGE RATION.
143?DIVIDE
144  PRTX
145  RCL6
146  RCL0          FIND THE AVERAGE GCE.
147?DIVIDE
148  PRTX
```

```
149 *LBLD
150  CLRG
151  P=S    (SWITCH PRIMARY AND SECONDARY REGISTERS)
152  SPC
153  RCL8
154  RCL4
155  RCL6
156  x
157  RCL9
158?DIVIDE
159  -
160  ENT
161  ENT
162  RCL4
163 X SQUARED (NO SYMBOL AVAILABLE)
164  RCL9
165?DIVIDE
166  RCL5
167  X=Y       (SWITCH X AND Y)
168  -
169?DIVIDE
170  STOB           WORK OUT THE CONDITION FACTOR AND
171  x              THE CONSTANTS a AND b FOR LINEAR
172  RCL6           REGRESSION.
173 X SQUARED
174  RCL9
175?DIVIDE
176  CHS
177  RCL7
178  +
179?DIVIDE
180  PRTX
181  RCL6
182  RCL4
183  RCLB
184  x
185  -
186  RCL9
187?DIVIDE
188  STOA
189  PRTX
190  RCLB
191  PRTX
192  RCLB
193  1/X
194  RCLA
195  0
196  X=Y       (SWITCH X AND Y)
197  -
198  x              USE LINEAR REGRESSION TO FIND THE
199  PRTX           MAINTENANCE RATION (WHEN THE GROWTH
200  STO1           IS ZERO).
201  0
202  STO0
203 *LBLe
204  1              COUNT THE SAMPLE NUMBER.
205  ST+0
206  SPC
207  R/S
208  STO2
209  RCL0
210  PRTX
211  RCL2
212  R/S
213  RCL1
214  -              NCE=GROWTH/(RATION - MAINTENANCE RATION)
215?DIVIDE
216  PRTX
217  GTOe
218 *LBLa
219  SF0
220  CF1            FOR ERROR CORRECTION SET FLAG ZERO.
221  GTOC
222  RTN
```

C-N COMPUTATION

This program calculates the carbon and nitrogen content in micrograms for a given dry weight or volume of organic or sedimentary material combusted in an elemental analyzer directly from the readout of a chart recorder (full scale = 1 mVolt). Sensitivity values (μV/μg), zero values (chart units, c.u.) and blank values (c.u.) for both carbon and nitrogen are stored at the beginning of the program. The option for dealing with dry weight of material (option 1) or filtered samples of known volume (option 2) is also determined at the beginning of the program. For each sample calculation, the appropriate read value (c.u.) and dry weight (μg) or volume (mL) are entered. Stored zero and blank values are subtracted from the entered read value (carbon or nitrogen) and the result multiplied by 10 to convert from chart units to μV. These values are divided by the appropriate sensitivity value to obtain μg carbon or nitrogen. The final output for each sample consists of a calculation of the percentage carbon content followed by a calculation of the percentage nitrogen content for option 1 or calculation of the concentration of carbon and nitrogen in μg/mL (equivalent to mg/L) for option 2. The program has the flexibility to allow changes to the zero and blank values of carbon and nitrogen during a run.

Equations pertaining to both carbon and nitrogen:

OPTION 1

$$\frac{[\text{Read value (c.u.)}-\text{Zero value (c.u.)}-\text{Blank value (c.u.)}] \times 10}{\text{Sensitivity } (\mu V/\mu g)} \text{ / Dry weight } (\mu g) \times 100 = \%C \text{ or } N$$

OPTION 2

$$\frac{[\text{Read value (c.u.)}-\text{Zero value (c.u.)}-\text{Blank value (c.u.)}] \times 10}{\text{Sensitivity } (\mu V/\mu g)} \text{ / Volume (mL)} = \mu g/mL \text{ C or N,}$$

where 1 chart unit (c.u.) = 10 microvolts (μV)
μg = micrograms
mL = milliliters

Sample problem

Given carbon sensitivity (CSen) = 20.87
nitrogen sensitivity (NSen) = 7.44

Weight	Nitrogen or carbon	Read (CRead, NRead)	Zero (CZero, NZero)	Blank (CBlank, NBlank)
2315	C	2580	219	17.5
	N	43.4	4.8	18.2
3455	C	3728	219	17.5
	N	52.6	4.8	18.2
4993	C	5631	219	17.5
	N	101.6	4.8	18.2
2427	C	2114	217	23.0
	N	41.4	4.7	17.8

Find total c.u., μg C, μg N, % N, and % C for each sample.

Input	Function		Display	Comments
	(1) D			
	or (2) F	E	0.000	(1) Option 1 or (2) Option 2
20.87		R/S	20.87	Enter carbon sensitivity
7.44		R/S	7.44	Enter nitrogen sensitivity
219		R/S	219	Enter carbon zero value
4.8		R/S	4.8	Enter nitrogen zero value
17.5		R/S	17.5	Enter carbon blank value
18.2		R/S	20.87	Enter nitrogen blank value
(a) 2580		R/S	PRINTED	Enter carbon read value
			20.87	Carbon sensitivity
			1	Sample number
			219	CZero
			17.5	CBlank
			2580	CRead
			2343.5	Total c.u.
			1122.904	µg C
2315		R/S	48.506	% C (0.485 for option 2)
43.4		R/S	7.44	NSen
			2	Sample number
			4.8	NZero
			18.2	NBlank
			43.4	NRead
			20.4	Total c.u.
			27.419	µg N
2315		R/S	1.184	% N (0.012 for option 2)

Samples 2 and 3 are worked out the same as sample 1. Go to (a) and enter the new CRead value (the rest is done for you - CSen, CZero, CBlank, NSen, NZero, and NBlank do not have to be re-entered).

In sample 4 the Zero, Blank and Read values have to be changed.

Input	Function		Display	Comments
	F	D		
217		R/S	217	Enter new carbon zero value
		C		
23		R/S	23	Enter new carbon blank value
	F	A	20.87	Returns you to the program (this could have been entered after the first R/S if only the CZero was to have been changed)
2114		R/S	PRINTED	Enter new carbon read value
			20.87	Carbon sensitivity
			7	Sample number
			217	New CZero
			23	New CBlank
			2114	CRead
			1874	Total c.u.
			897.44	µg C
2427		R/S	36.998	% C (0.370 for option 2)
		E	7.44	
17.8		R/S	17.8	Enter new nitrogen blank value
	F	C		
4.7		R/S	4.7	Enter new nitrogen zero value
		B	7.44	This returns you to the program (if only the blank value was to have changed [B] could have been typed after the first R/S)
41.4		R/S	PRINTED	Enter new nitrogen read value
			7.44	NSen
			8	Sample number
			4.7	New NZero
			17.8	New NBlank
			41.4	NRead
			18.9	Total c.u.
			25.403	µg N
2427			1.047	% N (0.010 for option 2)

	INSTRUCTIONS	INPUT	FUNCTION		DISPLAY
1	LOAD THE PROGRAM				PRINTED
2	FOR OPTION 1			D	
	FOR OPTION 2		F	E	
3	ENTER CARBON SENSITIVITY	CSEN		R/S	
4	ENTER NITROGEN SENSITIVITY	NSEN		R/S	
5	ENTER CARBON ZERO	CZERO		R/S	
6	ENTER NITROGEN ZERO	NZERO		R/S	
7	ENTER CARBON BLANK	CBLANK		R/S	
8	ENTER NITROGEN BLANK	NBLANK		R/S	
9	ENTER CARBON READ	CREAD		R/S	
					CSEN
					SAMPLE #
					CZERO
					CBLANK
					CREAD
					TOTAL c.u.
					µg C
10	ENTER SAMPLE WEIGHT	WEIGHT		R/S	%C
11	ENTER NITROGEN READ	NREAD		R/S	
					NSEN
					SAMPLE #
					NZERO
					NBLANK
					NREAD
					TOTAL c.u.
					µg N
12	ENTER SAMPLE WEIGHT	WEIGHT		R/S	%N
13	RETURN TO 9 AND ENTER THE NEXT CARBON READ.				
	TO CHANGE THE NBLANK WAIT UNTIL THE NEXT NREAD IS TO BE ENTERED THEN...			E	
	ENTER THE NEW NBLANK VALUE. B PUTS THE PROGRAM BACK TO STEP 11.	NBLANK		R/S B	
	TO CHANGE THE NZERO WAIT UNTIL THE NEXT NREAD IS TO BE ENTERED THEN ...		F	C	
	ENTER THE NEW NZERO RETURN TO STEP 11.	NZERO		R/S B	
	TO CHANGE THE CBLANK WAIT UNTIL THE NEXT CREAD IS TO BE ENTERED THEN...			C	
	ENTER THE NEW CBLANK RETURN TO STEP 9.	CBLANK	F	R/S A	
	TO CHANGE THE CZERO WAIT UNTIL THE NEXT CREAD IS TO BE ENTERED THEN...		F	D	
	ENTER THE NEW CZERO RETURN TO STEP 9.	CZERO	F	R/S A	
	THE ONLY WAY TO CHANGE THE CARBON OR NITROGEN SENSITIVITY IS TO RE-START THE PROGRAM.				

PROGRAM LISTING

```
001 *LBLA
002  0
003  STO0
004  R/S
005  STOB          STORE CARBON SENSITIVITY (CSEN) IN REGISTER B
006  R/S
007  STOC          STORE NITROGEN SENSITIVITY (NSEN) IN REGISTER C
008 *LBLd
009  R/S
010  STO4          STORE CARBON ZERO VALUE (CZERO) IN REGISTER 4
011 *LBLc
012  R/S
013  STO5          STORE NITROGEN ZERO VALUE (NZERO) IN REGISTER 5
014 *LBLC
015  R/S
016  STOD          STORE CARBON BLANK VALUE IN REGISTER D
017 *LBLE
018  R/S
019  STOE          STORE NITROGEN BLANK VALUE IN REGISTER E
020 *LBLa
021  RCL4
022  STO1
023  RCLD
024  STO2          INITIALIZE REGISTERS
025  RCLB
026  STO6
027 *LBL1
028  R/S
029  STO3          STORE CARBON OR NITROGEN READ VALUE IN REGISTER 3
030  SPC
031  RCL6          PRINT CARBON OR NITROGEN SENSITIVITY
032  PRTX
033  DSP0
034  1
035  ST+0          PRINT SAMPLE NUMBER (ODD FOR CARBON, EVEN
036  RCL0          FOR NITROGEN)
037  PRTX
038  DSP3
039  RCL1          PRINT ZERO VALUE
040  PRTX
041  RCL2          PRINT BLANK VALUE
042  PRTX
043  RCL3          PRINT READ VALUE
044  PRTX
045  RCL3
046  RCL2
047  -             CALCULATE AND PRINT TOTAL c.u.
048  RCL1          READ-BLANK-ZERO=TOTAL c.u.
049  -
050  PRTX
```

```
051  1
052  0
053  x
054  RCL6          CALCULATE AND PRINT N/C
055?DIVIDE        (TOTAL c.u. x 10)/SENSITIVITY=N/C
056  PRTX
057  R/S           ENTER SAMPLE WEIGHT
058?DIVIDE
059  F1?           SKIP IF OPTION D TAKEN
060  GTO2
061  1
062  0
063  0             CALCULATE AND PRINT THE %NC
064  x             %NC=(N/C)/SAMPLE WEIGHT x100%
065 *LBL2
066  PRTX
067 *LBLb
068  RCL0
069  2
070?DIVIDE          CHECK TO SEE WHETHER THE SAMPLE IS ODD
071  FRC           OR EVEN (CARBON OR NITROGEN).
072  X=0?
073  GTOa
074 *LBLB
075  RCL5
076  STO1
077  RCLE
078  STO2          INITIALIZE THE REGISTERS FOR A NITROGEN SAMPLE.
079  RCLC
080  STO6
081  GTO1
082 *LBLe
083  SF1           CHOOSE OPTION 1
084  GTOA
085 *LBLD
086  CF1           CHOOSE OPTION 2
087  GTOA
088  RTN
```

REFERENCES

Banse, K., and S. Mosher. 1980. Adult body mass and annual production/ biomass relationships of field populations. Ecol. Monogr. 50: 355-379.

Crisp, D. J. 1975. Secondary productivity in the sea, p. 71-89. In Productivity of World Ecosystems. National Academy of Sciences, Washington, USA.

Mills, E. L., and R. O. Fournier. 1979. Fish production and the marine ecosystems of the Scotian Shelf, eastern Canada. Mar. Biol. 54: 101-108.

Robertson, A. I. 1979. The relationship between annual production:biomass ratios and lifespans for marine macrobenthos. Oecologia (Biol.) 38: 193-202.

Schwinghamer, P. 1981. Size distributions of benthic organisms in the Bay of Fundy. Ph.D. Thesis, Dalhousie University.

Steele, J. H. 1974. The Structure of Marine Ecosystems. Harvard Univ. Press, Cambridge, U.S.A., 128 p.

Wildish, D. J., and D. Peer. 1981. Methods for estimating secondary production in marine Amphipoda. Can. J. Fish. Aquat. Sci. 38: 1019-1026.

Appendix 1. Modified HP-97 program for Biolumat ATP measurements incorporating correlation and regression analyses of standards, program modification of card supplied with the Biolumat ATP machine.

	Input	Function		Display	Comments
		F	B	PRINTED	Begin program. This will clear all the flags and registers.
			B		Sets the flags and registers used in the program.
(1)			E	Sample # Background Total count-background	Pressed when the sample goes in. It activates the machine and starts the program.
	Concentration		R/S	Total count-background	Every fifth sample, beginning at zero, is a standard. The program will stop to allow the entry of the concentration. Watch the sample number to tell when a standard is coming. If, when it comes the time to enter the concentration, you decide not to use this standard in the calculation of the curve instead of entering the concentration, enter a new unknown sample and go to (1).
	Multiple of 5	STO	0		If you want to do more standards without entering a multiple of 5 and store it in Register 0, enter four new samples and go to (1).
			C	R^2 a b	When all the samples are entered [C] gives the R^2, a, b values of the curve fitting.
	Unknown count	F	D	Unknown count concentration	Enter all your unknown counts to get the approximate concentrations.

Appendix 2. Growth/ration program for arithmetic linear model.

The only difference between this growth/ration program and the one on page 12 is that the formula used is different and uses:

$$\frac{\text{Final weight - Initial weight}}{\text{Initial weight x time}} \times 100\%.$$

The instructions for the program remain unchanged.

Appendix 3. Production ecology aide-memoire.

This program aids in recalling the formulas for calculating P:B which were published by 1983. The program also calculates P:B ratios based on wet weight or kilocalories (Kcal) and lifespan.

Equivalents to convert wet weight, g, to Kcal multiply by:

- 1.0 Steele (1974)
- 0.6 Mills and Fournier (1979) - refers to fresh weight

or - 0.5 Crisp (1975)

After entering the appropriate equivalent, type C. The equivalent may be changed by entering a new value and retyping C.

The following equations may be used by entering w, the wet weight[1] ($g.m^{-2}.yr^{-1}$), followed by the appropriate letter.

Letter	Equation	Applicability	Reference
D	$P:B=0.971w^{-0.167}$	Macrofauna + meiofauna	Schwinghamer (1981)
E	$P:B=0.112w^{-0.299}$	Meiofauna	"
f A	$P:B=0.536w^{-0.302}$	Macrofauna	"
B	$P:B=0.646w^{-0.37}$	Invertebrates, 5-20°C	Banse and Mosher (1980)
f B	$P:B=0.617w^{-0.39}$	Invertebrates without insects	"
A	$P:B=4.571\ \text{lifespan (yr)}^{-0.726}$	Macrofauna	Robertson (1979)

[1]Note that this does not apply when letter A is used and the entry here must be preceded by the lifespan in years.

Note that equations from Banse and Mosher (1980) and Robertson (1979) are given in a simpler form than they were originally presented by these authors.

Output

wet weight
Kcal
P:B

Note that for letter A the value given is the lifespan and not wet weight.

Data as Kcal may be converted to units of kilojoules (KJ) as follows:

Kcal x 4.18

by simply pressing fE.